Maple Commands for Chapter 14 (Stewart)

1 Graph the function

Graph $z = f(x, y)$. In this example, $z = \sqrt{9 - x^2 - y^2}$

```
f:=(x,y)->sqrt(9-x^2-y^2);
plot3d(f(x,y),x=-3..3,y=-3..3);
```

Compare that with the following:

```
z:=sqrt(9-x^2-y^2);
plot3d(z,x=-3..3,y=-3..3);
```

Exercise: Plot $z = \sin(xy)$, for $x \in [-10, 10]$ and $y \in [-10, 10]$. Change the color scheme to "Z Hue", and increase the number of points used to plot.

Note that when you click your mouse on the graph, you can spin it around. You can also change how the axes look, put on a legend, change the coloring, etc. Experiment with the menu options (with the black box around the figure- otherwise, you get the regular menu options).

2 Plot the level curves

Plot the level curves (also called contours) of the function $z = f(x, y)$. The Maple command is `contourplot`

```
with(plots):
z:=(x+y)/(sin(y)+2);
contourplot(z,x=-3..3,y=-3..3);
```

Now change the number of points used and see if the graph changes. We can also tell Maple which contourplots to graph. In this example, compare the plot of $f(x, y) = \sin^2(x) + \frac{1}{4}y^2$ with its contours at $1/10, 1/2, 1, 3$. Plot the default contours in three dimensions.

```
g:=(sin(x))^2+(1/4)*y^2;
plot3d(g,x=-5..5,y=-2..2);
contourplot(g,x=-5..5,y=-2..2,contours=[1/10,1/2,1,3]);
contourplot3d(g,x=-5..5,y=-2..2);
```

3 Multiple Limits.

EXAMPLE: Does $\lim_{(x,y)\to(0,0)} \frac{x^2 - y^2}{x^2 + y^2}$ exist?

```
limit( (x^2-y^2)/(x^2+y^2), {x=0, y=0} );
```

Note: If Maple cannot determine the value of the limit, the limit expression will be returned unevaluated. For example,

```
limit((sin(x^2)-sin(y^2))/(x-y), {x=0,y=0});
```

EXAMPLE: Use a graph of the function to determine if the following limit exists (try a contourplot, too).

$$\lim_{(x,y)\to(0,0)} \frac{2x^2 + 3xy + 4y^2}{3x^2 + 5y^2}$$

```
f:=(x,y)->(2*x^2+3*x*y+4*y^2)/(3*x^2+5*y^2);
plot3d(f(x,y),x=-1..1,y=-1..1);
```

4 Partial Derivatives:

4.1 Example:

If $f(x, y, z) = xe^{xy} \ln(z)$, compute all first partial derivatives and all second partials involving x and z.

```
restart;
f:=x*exp(x*y)*ln(z);
fx:=diff(f,x);
fy:=diff(f,y);
fz:=diff(f,z);
fxz:=diff(fx,z);
fzx:=diff(fz,x);
fyz:=diff(fy,z);
```

and so on.

4.2 Using the Definition:

Use the definition to compute $f_x(x, y, z)$, if

$$f(x, y, z) = \sqrt{x^2 + y^2 + z^2}$$

```
f:=(x,y,z)->sqrt(x^2+y^2+z^2);
DiffQuot:=(f(x+h,y,z)-f(x,y,z))/h;
fx=limit(DiffQuot,h=0);
```

5 Tangent Planes

Find the expression for the tangent plane, and plot the plane together with the function.

EXAMPLE: If $f(x, y) = xe^{xy}$, plot $z = f(x, y)$ together with the tangent plane to f at $x = 1, y = 1$.

First, rewrite the equation of the tangent plane:

$$z = z_0 + f_x(x_0, y_0)(x - x_0) + f_y(x_0, y_0)(y - y_0)$$

Using Maple:

```
f:=(x,y)->x*exp(x*y);
fx:=diff(f(x,y),x);
fy:=diff(f(x,y),y);
fx1:=subs({x=1,y=1},fx);
fy1:=subs({x=1,y=1},fy);
P:=f(1,1)+fx1*(x-1)+fy1*(y-1);
plot3d({f(x,y),P},x=-1..3,y=-1..3,view=0..5,axes='boxed');
```

6 The Chain Rule

(See Section 14.5, Example 5, p. 935 of Stewart's Calc)

If $u = x^4y + y^2z^3$, and $x = rse^t$, $y = rs^2e^{-t}$, and $z = r^2s\sin(t)$, find the value of $\frac{\partial u}{\partial s}$ when $r = 2, s = 1, t = 0$.

In Maple:

```
x:=r*s*exp(t);
y:=r*s^2*exp(-t);
z:=r^2*s*sin(t);
u:=x^4*y+y^2*z^3;
h:=diff(u,s);
h1:=subs({r=2,s=1,t=0},h);
evalf(h1);
```

7 Implicit Differentiation:

If $F(x, y) = 0$, then $\frac{dy}{dx} = \frac{-F_x}{F_y}$

EXAMPLE: Find y' if $x^3 + y^3 = 6xy$.

In Maple:

```
F:=x^3+y^3-6*x*y;
dydx:=diff(F,x)/diff(F,y);
```

8 The Gradient

Compute the Gradient, $\nabla f = [f_x, f_y, f_z]$

In Maple, if $f(x, y, z) = 3x^2 + 2yz$, then the gradient is:

```
with(linalg):
grad(3*x^2+2*y*z, vector([x,y,z]));
```

9 Contours and Gradients

Plot the contours of $f(x, y) = x^2 - y^2$, together with some gradient vectors (See Section 14.6, Figure 13, pg. 950)

```
with(plots):
A:=contourplot(x^2-y^2,x=-4..4,y=-4..4):
B:=gradplot(x^2-y^2,x=-4..4,y=-4..4,grid=[6,6],arrows='slim'):
display({A,B});
```